

Description

In this example, we demonstrate how to use `collect` to store the results of `anova` and how to use a predefined style to produce an ANOVA table with these results.

Remarks and examples

Below, we use data from the Second National Health and Nutrition Examination Survey (NHANES II) (McDowell et al. 1981). We wish to test whether the mean systolic blood pressure (`bpsystol`) is the same across males and females and across each category of race. To perform these tests, we use the `anova` command. We first create a new collection named `ex8` and then use the `collect` prefix to collect the results from `anova` and store them in the new collection.

```
. use https://www.stata-press.com/data/r19/nhanes21
(Second National Health and Nutrition Examination Survey)
. collect create ex8
(current collection is ex8)
. quietly: collect : anova bpsystol sex race
```

All results that `anova` returns in `e()` and `r()` will be collected. These results include the usual linear regression estimation results, sums of squares of the model terms, their degrees of freedom, F statistics, and p -values.

Stata ships with a predefined style named `anova` suitable for producing ANOVA tables after `anova` and `oneway`. Here we replace the default style with the `anova` style and call `collect preview` to report our ANOVA table.

```
. collect style use anova, replace
Collection: ex8
  Rows: term
Columns: result[SS df MS F p]
  Tables: cmdset
Table 1: 5 x 5
. collect preview
```

	SS	df	MS	F statistic	Prob>F
Model	48978.71	3	16326.24	30.24	<0.0001
sex	38166.29	1	38166.29	70.70	<0.0001
race	11316.64	2	5658.32	10.48	<0.0001
Residual	5585691.32	10,347	539.84		
Total	5634670.03	10,350	544.41		

Use `collect style` commands to make further style changes. Use the `collect label levels` command to make any necessary label changes. Once we are happy with our ANOVA table's style and labels, we can export it to another format with `collect export`.

□ Technical note

If you collect results from multiple calls to `anova` and use the above predefined style, the default layout will produce a separate table for each collected call to `anova`. The `cmdset` dimension indexes the collected estimation results, and the above predefined style specifies it in the layout to produce the separate tables.

Suppose we collected estimation results from a second ANOVA model.

```
. quietly: collect: anova bpsystol sex##race
```

To see how the estimation command results and `cmdset` levels are assigned, type

```
. collect layout (cmdset) (result[cmdline])
```

To restore the layout to show the above ANOVA tables, type

```
. collect style use anova, layout
```

If we want to publish the second model fit in our report, we can select it using `collect style autolevels` with the `cmdset` dimension without having to change `collect layout` by typing

```
. collect style autolevels cmdset 2, clear
. collect preview
```



Reference

McDowell, A., A. Engel, J. T. Massey, and K. Maurer. 1981. “Plan and operation of the Second National Health and Nutrition Examination Survey, 1976–1980”. In *Vital and Health Statistics*, ser. 1, no. 15. Hyattsville, MD: National Center for Health Statistics.

Also see

[TABLES] [collect style use](#) — Use collection styles from disk

[TABLES] [Predefined styles](#) — Predefined collection styles

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

